Étude de fonction (1)

La correction partielle (seule la partie *calculatoire* est développée) de l'exercice ci-dessous comprend quelques interactions avec MAXIMA et la représentation de fonctions.

1/ Soit g la fonction numérique de la variable réelle x définie par $g(x) = -\dfrac{1}{2} + \dfrac{x}{2\sqrt{x^2+1}}$.

Étudier la continuité et la dérivabilité de g ; en déduire les variations de g.

2/ Soit f la fonction de la variable réelle x définie par $f(x) = -\dfrac{x}{2} + 1 + \dfrac{1}{2}\sqrt{x^2+1}$.

Étudier la fonction f ; étudier la position de la courbe représentative de f, $\mathscr{C}_f$, par rapport à ses asymptotes, puis construire $\mathscr{C}_f$ dans un repère orthonormé $(O, \mathbf{i}, \mathbf{j})$ (unité : 2 cm).

3/ Déduire de l'étude précédente l'existence d'un intervalle I de $\mathbf{R}$, à préciser, tel que f permette de définir une bijection de $\mathbf{R}$ sur I. Vérifier que la bijection réciproque est telle que, pour tout x de I :

$$f^{-1}(x) = \frac{1}{4(x-1)} + 1 - x$$

Tracer la courbe $\mathscr{C}_{f^{-1}}$, représentant f^{-1} dans $(O, \vec{\imath}, j\vec{math})$.

```
> radexpand:all;
```

$$ALL$$

1/ Définition de g, dérivée et limites :

```
> g(x):=-1/2+x/2/sqrt(x^2+1);
```

```
g(x):=(-1)/2+x/2/SQRT(x^2+1);
```

```
> radcan(diff(g(x),x));
```

$$\frac{\sqrt{x^2+1}}{2\,x^4+4\,x^2+2}$$

La dérivée de g est manifestement positive sur **R**.

```
> limit(g(x),x,minf);
```

$$-1$$

```
> limit(g(x),x,inf);
```

$$0$$

g est donc strictement croissante sur **R**, elle varie de -1 à 0.

2/ Étude de f.

```
> f(x):=-x/2+1+1/2*sqrt(x^2+1);
```

```
f(x):=(-x)/2+1+1/2*SQRT(x^2+1);
```

```
> diff(f(x),x);
```

$$\frac{x}{2\sqrt{x^2+1}}-\frac{1}{2}$$

On constate l'égalité : $f'(x) = g(x)$, f' est donc négative sur **R**, la fonction f est décroissante sur **R**.

```
> limit(f(x),x,minf);
```

$$\infty$$

```
> limit(f(x),x,inf);
```

$$1$$

f varie décroit donc de $-\infty$ à 1. Nous recherchons maintenant une éventuelle asymptote à C_f au voisinage de $-\infty$.

```
> a:limit(f(x)/x,x,minf);
```

$$-1$$

```
> b:limit(f(x)-a*x,x,minf);
```

$$1$$

La droite d'équation $y = -x+1$ est asymptote vers $-\infty$. Pour déterminer la position relative de cette droite par rapport à la courbe, nous calculons la quantité $f(x) - (-x+1)$:

```
> d:radcan(f(x)-a*x-b);
```

$$\frac{\sqrt{x^2+1}+x}{2}$$

Il est assez facile de *voir* que cette quantité est positive. Pour confirmer, nous allons utiliser un développement.

Tout d'abord on se déplace sur un voisinage à droite de 0 :

```
> d:subst(-1/y,x,d);
```

$$\frac{\sqrt{\frac{1}{y^2}+1}-\frac{1}{y}}{2}$$

On développe ensuite...

```
> d:taylor(d,y,0,2);
```

$$+\frac{y}{4}+\cdots$$

Et on retourne au voisinage de $-\infty$.

```
> d:subst(-1/x,y,d);
```

$$-\frac{1}{4\,x}$$

C'est bien positif au voisinage de $-\infty$!

C_f admet une asymptote horizontale au voisinage de $+\infty$. Compte tenu des variations de f, cette asymptote est située en dessous de la courbe (aucun calcul à faire).

3/ *Réciproque* de f.

On définit h la fonction *prétendante* donnée par l'énoncé.

```
> h(x):=1/(4*(x-1))+1-x;
```

```
h(x):=1/(4*(x-1))+1-x;
```

On compose avec f à droite :

```
> i:h(f(x));
```

$$\frac{1}{4\left(\frac{\sqrt{x^2+1}}{2}-\frac{x}{2}\right)}-\frac{\sqrt{x^2+1}}{2}+\frac{x}{2}$$

Et on simplifie :

```
> radcan(i);
```

$$x$$

C'est le résultat attendu...

On compose maintenant avec f à gauche (pour vérifier) :

```
> i:f(h(x));
```

$$\frac{x-\frac{1}{4\,(x-1)}-1}{2}+\frac{\sqrt{\left(-x+\frac{1}{4\,(x-1)}+1\right)^2+1}}{2}+1$$

```
> radcan(i);
```

$$x$$

h est bien la réciproque de $f:\mathbf{R}\to]1,+\infty[$.


y
$y = f^{-1}(x)$
$y = -x + 1$
$y = f(x)$
$+1$
$y = 1$
x
$+1$